

Immobilienfinanzierung: Kreditvergleich mit unterschiedlicher Zinsbindungsdauer

NORBERT BRUNNER UND MANFRED KÜHLEITNER, WIEN

Zusammenfassung: Immobilienkredite werden überwiegend mit fester Zinsbindung mit Anschlusskredit zurückbezahlt. Dieser Artikel vergleicht anhand historischer Zinsdaten einjährige und zehnjährige Zinsbindung auf Vorteilhaftigkeit. Die Verwendung eines Tabellenkalkulationsprogramms ermöglicht einen elementaren Zugang bei gleichzeitiger realitätsnaher Verwendung großer Datenmengen.

1 Einleitung

Wenn eine Bank Geld verleiht, verlangt sie dafür Zinsen, mit denen sie ihre eigenen Kosten abdeckt (Finanzierung, Personal) und für das Risiko vorsorgt. Bei Krediten mit langfristiger Zinsbindung kommt das Opportunitätsrisiko dazu, dass der marktübliche Zinssatz steigt und die Bank auf höhere Einnahmen verzichtet. Deswegen sind die Zinsen für langfristige Zinsbindungen höher als für kurzfristige Bindungen. Viele Konsumenten bevorzugen dennoch die langfristige Bindung, weil sie Planungssicherheit für ihre Finanzen wollen. Wir fragen in dieser Arbeit: Waren in der Vergangenheit langfristige oder kurzfristige Zinsbindungen vorteilhafter? Wir beantworten diese Frage mit einer Simulation auf der Grundlage historischer Zinssätze.




Abb. 1: 12-Monats Euribor Werte in Prozent

Daten

Als Zinssätze verwenden wir den 12-Monats Euribor (Zinssatz zwischen Banken). Die historischen Euribor Zinssätze sind im Internet verfügbar, z. B. unter dem Link

www.ariva.de/euribor_12_monate/historische_kurse

Für Kredite an Privatpersonen werden Aufschläge auf den Euribor vereinbart. Für Immobilienkredite (ge-

ringer Aufschlag, weil die Immobilie eine Sicherheit für die Tilgung ist) entnehmen wir Clostermann und Seitz (2019) die Aufschläge von 1,1 % für einjährige Bindung und 1,5 % für zehnjährige Bindung. Da der minimale Euribor Wert – 0,399 % war, blieben die Kreditzinsen alle positiv. (Solange das der Fall ist, müssen in Österreich die negativen Euribor-Werte bei der Berechnung des Zinssatzes berücksichtigt werden; Oberster Gerichtshof am 21.03.2017 zu 10 Ob 13/17k.)

Die Daten aus dem Internet (ab 30.12.1998) werden als CSV Datei exportiert, in Excel geöffnet und nach dem Datum aufsteigend sortiert. Tabelle 1 ist ein Ausschnitt aus einem Tabellenblatt und Abbildung 1 plottet die Werte (Spalte B in Tabelle 1).

	A	B
1	Datum	Euribor
2	30.12.1998	3,215
3	31.12.1998	3,213
4	04.01.1999	3,209

Tab. 1: Historische Euribor Werte (Prozent).

2 Hintergrund Finanzmathematik

Tilgungspläne

Wir nehmen an, dass ein Kredit von € 100 000 und einem Zinssatz von 1 % in fünf Jahren zurückgezahlt werden soll. Dazu betrachten wir unterschiedliche Tilgungsarten. In allen Fällen begegnen uns stets dieselben Begriffe: Die Folge der Annuitäten besteht aus den vom Schuldner pro Jahr geleisteten Zahlungen (für Zinsen und Tilgung) und die neue Restschuld ist die alte Restschuld minus die Tilgung. Mit diesen Begriffen kann man folgende Tabelle in Excel aufstellen.

	A	B	C	D	E
1	Jahr	Restschuld	Zinsen	Tilgung	Annuität
2	0	100 000	=1%*B2		=C2+D2
3	=A2+1	=B2-D2			

Tab. 2: Tilgungsplan Grundkonzept

Endfälliger Kredit

Hier werden jährlich nur Zinsen bezahlt. Getilgt wird der Kredit am Ende der Laufzeit.

	A	B	C	D	E
1	Jahr	Restschuld	Zinsen (€)	Tilgung (€)	Annuität (€)
2	0	€ 100 000,00	€ 1 000,00	€ 0,00	€ 1 000,00
3	1	€ 100 000,00	€ 1 000,00	€ 0,00	€ 1 000,00
4	2	€ 100 000,00	€ 1 000,00	€ 0,00	€ 1 000,00
5	3	€ 100 000,00	€ 1 000,00	€ 0,00	€ 1 000,00
6	4	€ 100 000,00	€ 1 000,00	€ 100 000,00	€ 101 000,00
7	5	€ 0,00			

Tab. 3: Endfälliger Kredit. Zinsen werden bezahlt, aber keine Tilgung. Getilgt wird erst im letzten Jahr.

Die laufenden Kosten sind gering (Vorteil), aber am Ende der Laufzeit muss die gesamte Kredithöhe zurückbezahlt werden (Nachteil: hohes Risiko).

Einmalerlag am Ende der Laufzeit

Eine andere Möglichkeit ist es, über die gesamte Dauer weder Zinsen noch Tilgung zu bezahlen. Die gestundeten Zinsen sind eine Krediterhöhung (negative Tilgung). In D2 steht daher die Formel $=-C2$. Am Ende der Laufzeit wird der Kredit durch einen Einmalerlag (z. B. aus einer Lebensversicherung) getilgt. Hier sieht der vollständige Tilgungsplan so aus:

	A	B	C	D	E
1	Jahr	Restschuld	Zinsen (€)	Tilgung (€)	Annuität (€)
2	0	€ 100 000,00	€ 1 000,00	-€ 1 000,00	€ 0,00
3	1	€ 101 000,00	€ 1 010,00	-€ 1 010,00	€ 0,00
4	2	€ 102 010,00	€ 1 020,10	-€ 1 020,10	€ 0,00
5	3	€ 103 030,10	€ 1 030,30	-€ 1 030,30	€ 0,00
6	4	€ 104 060,40	€ 1 040,60	€ 104 060,40	€ 105 101,01
7	5	€ 0,00			

Tab. 4: Tilgungsplan für den Einmalerlag am Ende der Laufzeit. Während der Kreditlaufzeit werden weder Zinsen noch Tilgungen bezahlt.

Alternativ kann man die Zelle E6 berechnen als $€ 100 000 \cdot (1 + p)^5$ mit $p = 1\%$ (Zinseszinsrechnung).

Konstante Tilgungshöhe

Eine weitere Möglichkeit besteht darin, jährlich gleiche Tilgungsraten zu leisten, also € 20 000, plus die fälligen Zinsen. Unten ist der Tilgungsplan:

	A	B	C	D	E
1	Jahr	Restschuld	Zinsen (€)	Tilgung (€)	Annuität (€)
2	0	€ 100 000,00	€ 1 000,00	€ 20 000,00	€ 21 000,00
3	1	€ 80 000,00	€ 800,00	€ 20 000,00	€ 20 800,00
4	2	€ 60 000,00	€ 600,00	€ 20 000,00	€ 20 600,00
5	3	€ 40 000,00	€ 400,00	€ 20 000,00	€ 20 400,00
6	4	€ 20 000,00	€ 200,00	€ 20 000,00	€ 20 200,00
7	5	€ 0,00			

Tab. 5: Konstante Tilgung. Die laufenden Zinsen werden vollständig bezahlt.

Aus der Tabelle sieht man die gesamte jährliche Belastung. Die Annuität ist am Anfang hoch, da die fälligen Zinsen bei hoher Restschuld hoch sind. Sie nehmen im Laufe der Tilgung ab, da die fälligen Zinsen mit Verringerung der Restschuld immer niedriger werden.

Konstante Annuität

Eine häufig genutzte Möglichkeit, besteht darin, innerhalb der gesamten Laufzeit jährlich gleichbleibende Leistungen zu erbringen. Wir müssen mit Hilfe eines Tilgungsplans die jährliche Rückzahlungsrate (Annuität) berechnen. Wir lösen die Aufgabe in Excel mit folgender Tabelle.

	A	B	C	D	E
1	Jahr	Restschuld	Zinsen (€)	Tilgung (€)	Annuität (€)
2	0	€ 100 000,00	€ 1 000,00	€ 19 603,98	€ 20 603,98
3	1	€ 80 396,02	€ 803,96	€ 19 800,02	€ 20 603,98
4	2	€ 60 596,00	€ 605,96	€ 19 998,02	€ 20 603,98
5	3	€ 40 597,98	€ 405,98	€ 20 198,00	€ 20 603,98
6	4	€ 20 399,98	€ 204,00	€ 20 399,98	€ 20 603,98
7	5	€ 0,00			

Tab. 6: Annuitätenkredit, mit konstanter Annuität.

Die Annuität soll konstant sein. Wir wissen allerdings nicht wie hoch. Daher schreiben wir in die Zelle E2 den Wert € 20 000 (Startwert für die Suche nach der unbekanntenen Annuität). In die Zelle E3 kommt die Formel $=E2$. Diese wird nach unten kopiert. Damit haben alle Annuitäten denselben Wert, der in E2 steht. Für die Tilgung schreiben wir in Zelle D2 die Formel $=E2-C2$. Diese Formel wird ebenfalls nach unten kopiert. Nun verwendet man die Zielwertsuche mit folgenden Einstellungen: Die Zielzelle B7 (Restschuld nach 5 Jahren) soll Null sein; die veränderliche Größe ist die Zelle E2.

Bemerkung für Lehrer: In der Schule wird nur diese Tilgungsmöglichkeit gerechnet. Mathematisch steckt die Rentenrechnung auf der Grundlage der geometrischen Reihe dahinter. Die Restschuld $R(n)$ im Jahr n berechnet man mit der linearen Rekursion: $R(0) = K = 100 000$ und $R(n+1) = (1+p) \cdot R(n) - a$ mit $p = 1\%$ und der unbekanntenen Annuität a . Gesucht ist eine Lösung a der Gleichung $R(5) = 0$. Die analytischen Lösungen (z. B. Mathematica) sind für die Rekursion $R(n) = (a + (p \cdot K - a) \cdot (1+p)^n) / p$ und für die lineare Gleichung $R(n) = 0$

$$a = K \cdot \frac{p \cdot (1+p)^n}{(1+p)^n - 1}$$

In Excel ist dies als Funktion RMZ() programmiert („regelmäßige Zahlung“). Die Annuität a berechnet

sich mit dem Befehl =RMZ(1%;5;-100 000;0) aus dem Zinssatz, der Laufzeit und der Kredithöhe. Die „0“ steht für nachschüssige Zahlungen. Da Excel kaufmännisch rechnet, wird die Schuld mit negativem Vorzeichen eingegeben.

	A	B	C	D	E
1	Jahr	Restschuld	Zinsen (€)	Tilgung (€)	Annuität (€)
2	0	€ 100 000,00	€ 1 000,00	€ 5 000,00	€ 6 000,00
3	1	€ 95 000,00	€ 950,00	€ 5 050,00	€ 6 000,00
4	2	€ 89 950,00	€ 899,50	€ 5 100,50	€ 6 000,00
5	3	€ 84 849,50	€ 848,50	€ 5 151,51	€ 6 000,00
6	4	€ 79 698,00	€ 796,98	€ 5 203,02	€ 6 000,00
7	5	€ 74 494,97	€ 744,95	€ 5 255,05	€ 6 000,00
8	6	€ 69 239,92	€ 692,40	€ 5 307,60	€ 6 000,00
9	7	€ 63 932,32	€ 639,32	€ 5 360,68	€ 6 000,00
10	8	€ 58 571,65	€ 585,72	€ 5 414,28	€ 6 000,00
11	9	€ 53 157,36	€ 531,57	€ 5 468,43	€ 6 000,00
12	10	€ 47 688,94	€ 476,89	€ 5 523,11	€ 6 000,00
13	11	€ 42 165,83	€ 421,66	€ 5 578,34	€ 6 000,00
14	12	€ 36 587,48	€ 365,87	€ 5 634,13	€ 6 000,00
15	13	€ 30 953,36	€ 309,53	€ 5 690,47	€ 6 000,00
16	14	€ 25 262,89	€ 252,63	€ 5 747,37	€ 6 000,00
17	15	€ 19 515,52	€ 195,16	€ 5 804,84	€ 6 000,00
18	16	€ 13 710,68	€ 137,11	€ 5 862,89	€ 6 000,00
19	17	€ 7 847,78	€ 78,48	€ 5 921,52	€ 6 000,00
20	18	€ 1 926,26	€ 19,26	€ 1 926,26	€ 1 945,52
21	18	€ 0,00			

Tab. 7: Prozentannuität. Die erste Tilgung ist ein gewisser Prozentsatz der Kredithöhe. Danach bleibt die Annuität bis auf die letzte Rate konstant. Laufzeit und letzte Zahlung ergeben sich aus dem Tilgungsplan.

Variable Laufzeit

Eine für Schuldner angenehme Möglichkeit, einen Kredit zurückzubezahlen ergibt sich so: Der Zinssatz betrage 1 % und aufgrund des regelmäßigen Einkommens kann man 5 % tilgen. Somit beträgt die erste Annuität 6 % der Anfangskredithöhe. Im weiteren Verlauf soll die Annuität konstant bei diesem Wert bleiben, bis auf die letzte Zahlung zur Tilgung der Restschuld. Mit dem Tilgungsplan wie in der Tabelle 7 berechnet die Tabelle 8 die Laufzeit und die letzte Zahlung. In diesem Fall ist die Annuität bekannt. Daher kann man die offene Restschuld nach 1, 2, 3, ... Jahren (Werte in Spalte B) berechnen.

Bemerkung: In Excel kann man für diese Rechnung die Funktion „zukünftiger Wert“ (Zeitwert) ZW() verwenden; so liefert =ZW(1%;6;6 000;-100 000;0) den Wert in Zelle B8. Die Variablen für ZW sind: Zinssatz, Anzahl der geleisteten Raten, Kredithöhe und „0“ für Rückzahlungen am Ende des Jahres.

Konstante Annuität mit Restschuldausgleich

Eine weitere Variante besteht darin, bis auf die letzte Rate jährlich gleichbleibende Leistungen zu erbringen, und die dann noch offene Restschuld mit der letzten Rate zu begleichen. Die Berechnung erfolgt wie in Tabelle 6, wobei wir in E6 die Formel B6+C6 für die letzte Rate eingeben. Wir nehmen an, dass € 22 000/Jahr bezahlt werden.

	A	B	C	D	E
1	Jahr	Restschuld	Zinsen (€)	Tilgung (€)	Annuität (€)
2	0	€ 100 000,00	€ 1 000,00	€ 21 000,00	€ 22 000,00
3	1	€ 79 000,00	€ 790,00	€ 21 210,00	€ 22 000,00
4	2	€ 57 790,00	€ 577,90	€ 21 422,10	€ 22 000,00
5	3	€ 36 367,90	€ 363,68	€ 21 636,32	€ 22 000,00
6	4	€ 14 731,58	€ 147,32	€ 14 731,58	€ 14 878,89
7	5	€ 0,00			

Tab. 8: Bis auf das letzte Jahr konstante Annuität und Zahlung der Restschuld mit der letzten Rate.

Bemerkung: Abhängig von Einkommen und zukünftigen Lebensverhältnissen (Erbschaft, Ausbezahlung einer Lebensversicherung) kann es sinnvoll sein, dass man einen Kredit vorzeitig zurückzahlt oder Sonderzahlungen tätigt. Auch Gebühren, Verzugs- oder Überziehungszinsen muss man sich mit der Bank ausmachen. All dies liefert weitere Kreditkonditionen.

3 Kreditvergleich

Wir haben gesehen, dass sich ein Kredit auf verschiedene Weisen zurückzahlen lässt. Wie also soll man kurzfristige und langfristige Kredite vergleichen? Wir sind von der häufigsten Form der Kredite ausgegangen, wo die Annuität während der Laufzeit konstant bleibt. Daher haben wir die konstante Annuität mit Restschuldausgleich als Rückzahlungsvariante gewählt.

Wir vergleichen zwei Kredite in der Höhe von € 100 000 und einer jährlichen Rückzahlung von jeweils € 6 000 über einen Zeitraum von zehn Jahren. Dabei hat ein Kredit eine zehnjährige Zinsbindung und der andere Kredit nur eine einjährige Zinsbindung, d. h. nach einem Jahr muss die Zinshöhe aktualisiert werden.

Der Vergleichszeitraum ergibt sich daraus, dass die meisten Zinsbindungen zehn Jahre betragen; noch länger macht keinen Sinn, weil sonst die Aufschläge zu hoch sind. Nach zehn Jahren ist eine Immobilie im Normalfall aber noch nicht abbezahlt, sondern der verbleibende Kredit muss umgeschuldet und die Zin-

sen neu ausgehandelt werden. Wir vergleichen daher die geleisteten Rückzahlungen (die sind in beiden Fällen € 60 000) und den ausstehenden Rest (als Restschuldzahlung) der zwei Kredite über einen Zeitraum von zehn Jahren.

Zum Vergleich wollen wir für einen beliebigen Tag untersuchen, welcher der beiden an diesem Tag aufgenommenen Kredite besser war, der mit kurzfristiger oder der mit langfristiger Bindung. Das hängt vom Zinssatz und dessen Veränderung ab. Deshalb notieren wir den Euribor-Wert plus Aufschlag (1,5 % bzw. 1,1 % bei lang- bzw. kurzfristiger Bindung) bei Kreditabschluss (Spalte H von Tabelle 9). Beim einjährig gebundenen Kredit schreiben wir zusätzlich die Zinssätze jeweils ein Jahr später an (Spalte H von Tabelle 11).

	C	D	E	F	G	H
1	Tag	Monat	Jahr	Wunschtag	Bank-Tag	Zinssatz
2	3	1	1999	03.01.1999	31.12.1998	0,04713

Tab. 9: Ergänzung von Tabelle 1 zur Berechnung des langfristigen Zinssatzes mit dem aktuellen Euribor-Wert.

Wir tragen in C2:E2 ein Datum (Tag-Monat-Jahr) ein, an dem wir einen Kredit aufnehmen und in Folge die Annuitäten erbringen wollen. In F2 steht das Datum: =VERKETTEN(C2;" ";D2;" ";E2)*1, formatiert als „Datum kurz“. Da dieser Tag kein Bank-Tag sein muss, berechnen wir in G2 mit =SVERWEIS(F2;A:B;1) jenes Datum aus Spalte A das unmittelbar vor oder gleich dem Wunschdatum ist. Das ist das Datum des letzten Bank-Tages vor dem gewünschten Datum, nach dem sich der Zinssatz richtet. Für diesen Tag gibt es einen Euribor-Wert, =SVERWEIS(F2;A:B;2). Aus ihm berechnen wir in H2 den Zinssatz bei zehnjähriger Bindung als: =(SVERWEIS(F2;A:B;2)+1,5)/100 (formatiert als Standard).

	I	J	K	L	M
1	Jahr	Restschuld	Zinsen	Tilgung	Annuität
2	0	100 000	=J2*\$H\$2	=M2-K2	6 000
3	1	=J2-L2	=J3*\$H\$2	=M3-K3	6 000
...	...				
12		=J11-L11			

Tab. 10: Ergänzung von Tabelle 1 um den Tilgungsplan für einen langfristig gebundenen Kredit

Mit dem Zinssatz aus H2 erstellen wir jetzt in den Spalten I bis M einen Tilgungsplan für den zehn Jahre gebundenen Kredit wie in Tabelle 10, wobei Zeile 3 bis in Zeile 11 kopiert wird.

Beim einjährig gebundenen Kredit variiert der Zinssatz jedes Jahr. Wir stellen dem Tilgungsplan eine Liste der jeweils aktuellen Zinssätze wie in Tabelle 11 voran. In C15:D15 steht das gleiche Wunsch-Datum für den Kreditbeginn und die Rückzahlungen, wie in C2:D2 für den langfristigen Kredit. Die Formeln in F2:G2 werden in die Zeile 15 kopiert und in H2 wird der Zinssatz für das erste Jahr aus dem Euribor des angegebenen Bank-Tages berechnet als: =(SVERWEIS(F15;A:B;2)+1,1)/100.

	C	D	E	F	G	H
14	Tag	Monat	Jahr	Wunschtag	Bank-Tag	Zinssatz
15	=C2	=D2	=E3	03.01.1999	31.12.1998	0,04713
16	3	1	=E15+1	03.01.2000	03.01.2000	0,04985

Tab. 11: Ergänzung von Tabelle 1 zur Auswahl von Euribor-Werten für kurzfristige Zinssätze

Nach dem Ablauf des Jahres wird der Zinssatz für die Restschuld neu festgesetzt: Das Datum dafür ist das alte Wunschdatum, nur 1 Jahr später: In C16 und D16 stehen die Formeln =C15 und =D15, und in E16 schreiben wir die Formel =1+E15. Ein Problem sind jedoch Schaltjahre: Der 29.2.2004 ist ein zulässiges Wunschdatum, doch im Folgejahr erscheint in F16 eine Fehlermeldung. Wir ändern daher in C16 das Datum 29.02. ab zu 28.02.; Eingabe =WENN(UND(C15=29;D15=2);28;C15). Wir kopieren nun die Zeile 16 bis in Zeile 24.

Tabelle 12 für den Tilgungsplan des einjährig gebundenen Kredits sieht fast gleich aus, wie Tabelle 10. Der Unterschied sind die Zinssätze: Statt des festen Zinssatzes in Zelle \$H\$2 (die \$-Zeichen fixieren den Bezug beim Kopieren) wird nun in jeder Zeile der aus dem aktuellen Euribor-Wert nach Ablauf eines Jahres berechnete Zinssatz verwendet.

	I	J	K	L	M
14	Jahr	Restschuld	Zinsen	Tilgung	Annuität
15	0	100 000	=J15*H15	=M15-K15	6 000
16	1	=J15-L15	=J3*H16	=M16-K16	6 000
...	...				
25		=J24-L24			

Tab. 12: Tilgungsplan für einen kurzfristig gebundenen Kredit, basierend auf den Euribor-Werten

Die Restschuld nach zehn Jahren lesen wir beim langfristig gebundenen Kredit in Zelle J12 ab und beim kurzfristig gebundenen Kredit in Zelle J25. Beide Kredite werden am selben Bank-Tag abgeschlossen. Je nachdem welcher Wert kleiner ist, erkennt man, welcher Kredit günstiger ist.

Bemerkung: Die Vergleichbarkeit der beiden Kredite ist deswegen gegeben, weil die Zahlungsflüsse bis auf die Restschuld gleich sind. Bei anderen Tilgungsplänen muss man die Zahlungsflüsse vor dem Vergleich diskontieren. (Der Grund ist, dass es für den Kreditnehmer mehr Nutzen hat, wenn er sich heute € 1 000 erspart, als wenn er sich in 10 Jahren die, um einen Ausgleich der Inflation erhöhten, € 1 000 ersparen kann.)

4 Umsetzung im Unterricht

Der klassische Mathematik-Unterricht kommt in der öffentlichen Diskussion weltweit immer mehr unter Kritik. „Education is not working to elevate society with modern computation, data science and AI. Instead, students are subjugated to compete with what computers do best, and lose“ (Wolfram 2020). Im Hinblick auf die Digitalisierung des Unterrichts bietet es sich daher an, Excel im Mathematik Unterricht frühzeitig einzubauen. Bei Schulen mit wirtschaftlichen Schwerpunkten bereitet dies auf das numerische Lösen mathematischer Probleme mit Excel beim wirtschaftswissenschaftlichen Studium vor. Business-Schools (auch Wirtschaftsuniversität Wien sowie wirtschaftsnahe Studien an der Universität für Bodenkultur) verwenden diesen Zugang schon lange für die einführenden Lehrveranstaltungen in Mathematik (vgl. Neuwirth und Arganbright 2004).

Die hier vorgestellten Tabellenblätter lassen sich im Unterricht unmittelbar einsetzen. Das Thema Immobilienkredit ist vielen Kindern aus der Familie bekannt (Wohnungskauf, Hausbau). Für den Vergleich von Krediten kann z. B. jedes Kind in C2:E2 ein anderes Datum einsetzen. In der Klasse kann dann aufgrund dieser Stichprobe verglichen werden, welcher Kredit günstiger war.

Warum war ein kurzfristig gebundener Kredit in der Vergangenheit meist günstiger? Die Schüler(innen) sollen dazu ein Datum finden, an dem der langfristig gebundene Kredit vorteilhafter war. Dies ist wie eine Suche nach der Nadel im Heuhaufen, da der langfristig-gebundene Kredit nur in 7 % der Fälle günstiger war. Wenn man in der Klasse sucht, sollte jeder zwei zufällige Daten testen (bei 21 Schülern 95 % Chance, solch ein Datum zu finden).

Will der Einzelne bei der Suche nicht einfach durch Versuch und Irrtum alle Euribor-Werte durchgehen, so muss man sich im Vorfeld etwas dazu überlegen. Wenn der Euribor konstant bliebe, oder sogar sinkt, so wäre der kurzfristige Kredit günstiger als der langfristige, weil der Aufschlag geringer ist. Interessant sind somit jene Zeiten, in denen die Zinsen über ei-

nen längeren Zeitraum gestiegen sind und der Zinsanstieg 0,4 % übertrifft (Unterschied zwischen kurz- und langfristigem Zinsaufschlag). Die Suche sollte sich daher an Abbildung 1 orientieren und Phasen mit langfristigem Zinsanstieg identifizieren.

In konkreten Kreditverträgen kann der Zinssatz zum Abschluss des Kredits bzw. bei der Neufestsetzung nach einem Jahr auch aus einem gewichteten Mittelwert mehrerer Euribor-Werte errechnet sein. Die Tabellen 9 und 11 können leicht entsprechend abgewandelt werden.

Als weitere Aufgabenstellung bietet sich die analoge Analyse des Einmalerlags am Ende der Laufzeit an.

5 Weitere Ergebnisse

Mit einem Makro kann man den Vergleich für alle 2831 Tage zwischen 30.12.1998 bis 20.01.2010 durchführen und die Ergebnisse in der Tabelle abspeichern. (Das Enddatum ergibt sich daraus, dass der 20.01.2020 das letzte Datum unseres Datensatzes war und daher die kurzfristig gebundenen Kredite mindestens zehn Jahre vorher starten mussten, um sie zu beurteilen.) Die Tabelle 13 fasst die Ergebnisse zusammen.

Wir schließen daraus: Sowohl die Restschuld als auch die Abhängigkeit vom Zeitpunkt der Kreditaufnahme (Streuung) waren bei einjähriger Zinsbindung im Mittel vorteilhaft gegenüber zehnjähriger Zinsbindung. In 2 625 Fällen war der langfristig gebundene Kredit teurer, und zwar um bis zu € 53 262 (in 50 % der Fälle mindestens um € 11 490 teurer) und nur in 206 Fällen (7 %) war dieser Kredit günstiger, allerdings um höchstens € 5 340.

	langfristig	kurzfristig	lang minus kurz
Minimum	62899	54384	- 5340
Maximum	114193	84389	53262
Mittelwert	86170	71290	14879
Median	84964	71899	11490
Standardabweichung	13431	8051	14014

Tab. 13: Statistische Kennzahlen für 2831 Restschulden bei lang- und kurzfristiger Bindung

Diese Analyse beruht auf dem historischen Verlauf eines stochastischen Prozesses (Euribor). Ob in Zukunft kurzfristige Bindungen weiterhin günstiger sind, bleibt abzuwarten. Die um 0,4 % höhere Differenz (Spread) des langfristig gebundenen Zinssatzes zum Euribor (im Vergleich zum kurzfristig gebundenen Kredit) ist nämlich aus der Sicht der Banken eine Versicherungsprämie für das Zinsrisiko. Falls der langfristig gebundene Kredit trotz Versicherungsprä-

mie im Durchschnitt günstiger wird als der kurzfristig gebundene Kredit, hätten die Banken zu wenig Prämie für dieses Risiko verlangt. (Wenn der langfristig gebundene Kredit mit Prämie im Durchschnitt gleich viel kostet, wie der kurzfristig gebundene Kredit, dann bezahlen die Kunden als Kollektiv mit ihrer Prämie das Risiko selbst. Tatsächlich sind viele Kunden risiko-avers. Sie sind daher bereit, für die Risikominderung noch mehr zu bezahlen.) Es wäre daher bei einer derartigen Zinsentwicklung zu erwarten, dass die Banken mit einem höheren Risikoaufschlag auf langfristig gebundene Kredite reagieren (oder keine langfristige Bindung anbieten, wenn sie das Risiko selbst noch nicht abschätzen können). Unter den neuen Konditionen wären kurzfristig gebundene Kredite dann wieder günstiger, aber auch mit einem Risiko behaftet.

Bemerkung für Lehrer: Für den automatisierten Vergleich mit einem Makro kopiert man das in Abschnitt 4 erstellte Tabellenblatt und verändert es wie folgt: Die Formel in Zelle G2 wird entfernt. Stattdessen werden dort vom Makro nacheinander Werte aus Spalte A eingetragen. In C2:E2 schreibt man dann die Formeln: =TAG(G2), =MONAT(G2), =JAHR(G2) und in F2 bleibt die Formel =VERKETTEN(C2;“.“;D2;“.“;E2)*1.

In Tabelle 14 wird ein einfaches Makro erklärt, das nacheinander die Datumswerte von Spalte A in Zelle G2 einsetzt. Das Tabellenblatt berechnet dann die Restschulden für den lang- und kurzfristig gebundenen Kredit, der an diesem Tag beginnt. Diese Werte (aus J12 und J25) werden in die Spalten V und W kopiert.

In den Zellen V1 und W1 kann man Beschriftungen eintragen (Restschuld-lang, Restschuld-kurz) und in Spalte U kann man zur besseren Orientierung eine Kopie von Spalte A einfügen.

Sub makro1()	Bezeichnung des Makros
For n = 2 To 2832	Beginn der Schleife
Cells(2, 7).Value = Cells(n, 1).Value	Kopie des Datums von Zeile n in Spalte A nach G2
Cells(n, 22).Value = Cells(12, 10).Value	Kopie des Werts von J12 in Zeile n von Spalte V
Cells(n, 23).Value = Cells(25, 10).Value	Kopie des Werts von J25 in Zeile n von Spalte W
Next n	Ende der Schleife
End Sub	Ende des Programms

Tab. 14: Makro für den automatisierten Kreditvergleich.

Literatur

- Clostermann, J., & Seitz, F. (2019): Effektivverzinsung und Volatilität bei Finanzierung mit Zinsbindung und variable Zinsen. Eine empirische Untersuchung für Deutschland. *Zeitschrift für Immobilienökonomie* 6(1), 29–46.
- Neuwirth, E., & Arganbright, D. (2004): *The Active Modeler: Mathematical Modeling with Microsoft Excel*. Thomson Brooks/Cde, Belmont.
- Wolfram, C. (2020): *The Math(s) Fix: An Education Blueprint for the AI Age*. Wolfram Media.

Anschrift der Verfasser

Norbert Brunner und Manfred Kühleitner
 Institut für Mathematik, DIBB
 Universität für Bodenkultur
 Gregor Mendel Strasse 33, A-1180 Wien
 Norbert.Brunner@BOKU.ac.at,
 Manfred.Kuehleitner@BOKU.ac.at